Precision Digital Sine-Wave Generation with the TMS32010

APPLICATION REPORT: SPRA007

Author: Domingo Garcia
Digital Signal Processing – Semiconductor Group

Digital Signal Processing Solutions
1989

TEXAS INSTRUMENTS

IMPORTANT NOTICE

Texas Instruments (TI) reserves the right to make changes to its products or to discontinue any semiconductor product or service without notice, and advises its customers to obtain the latest version of relevant information to verify, before placing orders, that the information being relied on is current.

TI warrants performance of its semiconductor products and related software to the specifications applicable at the time of sale in accordance with TI's standard warranty. Testing and other quality control techniques are utilized to the extent TI deems necessary to support this warranty. Specific testing of all parameters of each device is not necessarily performed, except those mandated by government requirements.

Certain application using semiconductor products may involve potential risks of death, personal injury, or severe property or environmental damage ("Critical Applications").

TI SEMICONDUCTOR PRODUCTS ARE NOT DESIGNED, INTENDED, AUTHORIZED, OR WARRANTED TO BE SUITABLE FOR USE IN LIFE-SUPPORT APPLICATIONS, DEVICES OR SYSTEMS OR OTHER CRITICAL APPLICATIONS.

Inclusion of TI products in such applications is understood to be fully at the risk of the customer. Use of TI products in such applications requires the written approval of an appropriate TI officer. Questions concerning potential risk applications should be directed to TI through a local SC sales office.

In order to minimize risks associated with the customer's applications, adequate design and operating safeguards should be provided by the customer to minimize inherent or procedural hazards.

TI assumes no liability for applications assistance, customer product design, software performance, or infringement of patents or services described herein. Nor does TI warrant or represent that any license, either express or implied, is granted under any patent right, copyright, mask work right, or other intellectual property right of TI covering or relating to any combination, machine, or process in which such semiconductor products or services might be or are used.

TRADEMARKS

CONTACT INFORMATION

US TMS320 HOTLINE	(281) 274-2320
US TMS320 FAX	(281) 274-2324
US TMS320 BBS	(281) 274-2323
US TMS320 email	dsph@ti.com

Precision Digital Sine-Wave Generation with the TMS32010

Abstract

This report presents two methods of sine-wave generation. The first method is a fast direct table lookup scheme suitable for applications where speech is critical. The second approach, an enhancement of the first, includes linear interpolation to provide higher accurate waveforms.

Product Support on the World Wide Web

Our World Wide Web site at www.ti.com contains the most up to date product information, revisions, and additions. Users registering with TI&ME can build custom information pages and receive new product updates automatically via email.

INTRODUCTION

Sine-wave generators are fundamental building blocks of signal processing systems which are used in diverse applications, such as communication, instrumentation, and control. In the past, engineers usually designed these oscillators with analog circuitry. Now, however, new high-speed digital signal processors like the TMS32010 present designers with an alternative that in many cases is superior. The TMS32010 provides the speed and accuracy to produce stable, low-distortion sine waves over a wide range of frequencies.

This application report describes two different methods for implementing a digital sine wave generator using the TMS32010. The first method is a fast direct table lookup scheme suitable for applications not requiring extreme accuracy. The second approach, an enhancement of the first, includes linear interpolation to provide sine waveforms with a minimum of harmonic distortion.

DIRECT TABLE LOOKUP METHOD

The first algorithm is a simple, fast table lookup scheme. The sine values for N angles which are uniformly spaced around the unit circle are stored in a table which has the following format:

INDEX	ANGLE	SINE TABLE
0	0 X 360°/N	S[0] = sin(0°/N)
1	1 X 360°/N	S[1] = sin(360°/N)
2	2 X 360°/N	S[2] = sin(720°/N)
.	.	.
.	.	.
.	.	.
N-2	(N-2) X 360°/N	S[N-2] = sin((N-2) X 360°/N)
N-1	(N-1) X 360°/N	S[N-1] = sin((N-1) X 360°/N)

A sine wave is generated by stepping through the table at a constant rate (in effect, moving counterclockwise around the unit circle), wrapping around at the end of the table whenever 360° is exceeded. Using the table index as the angle parameter and DELTA as the step size, this lookup method generates the sequence:

S[mod(k X DELTA,N)]   for k = 1, 2, 3, 4, ...

where mod(a,b) = remainder of the division a/b when this quotient is computed as an integer [e.g., mod(22.34,5) = 2.34)]

The 'mod' operator provides the wraparound at the end of the table. Figure 1 illustrates this algorithm.



Figure 1. Direct Table Lookup

The sampled waveform generated is only an approximation to a sampled sinusoid. In general, the longer the table is the more resolution it provides, and consequently, the closer the approximation will be.

The frequency, f, of the sine wave depends on two factors:

(1) The time interval between successive samples, i.e., the sampling interval, t

(2) The step size, DELTA

f is given by the equation:

$$f = \frac{DELTA}{t \times N} \text{ [Hz]} \quad \text{where t is expressed in seconds}$$

Note that to satisfy the Nyquist criterion there must be at least two samples generated each sinusoid period. This requires that DELTA $\leq$ N/2.

In Figure 1, N = 8 and DELTA = 0.6. If, for instance, eight samples are generated each millisecond, then t = 0.000125 seconds and

$$f = \frac{0.6}{8 \times 0.000125} \text{ Hz} = 600 \text{ Hz}$$

TMS32010 Implementation

This section describes the concise TMS32010 subroutine, given in Appendix B, which implements the table lookup scheme based on a sine table with 128 entries. Each time this subroutine is called, the next sample point is calculated. This subroutine uses:

(1) 138 (= 128 + 10) words of program memory space (128 words for sine table storage and 10 words for program memory)

(2) 6 words in data memory as working registers

If this program is used as a subroutine, each sample can be computed in 3.0 microseconds. However, if the code is inserted directly in line with the code of a master program, avoiding the overhead of a subroutine, a sample can be computed in 2.2 microseconds.

The values in the sine table are all scaled. The decimal values, +1.0 and −1.0, are represented by the two's complement hexadecimal values 4000 and C000, respectively. All other values are scaled and rounded to the closest hexadecimal number. Rounding is used, rather than truncation, to avoid adding unnecessary distortion.

The 16-bit data memory location 'ALPHA' serves as a modulo 128 counter which cycles through the sine table to select the sample points. ALPHA is regarded as having an integer and fractional part with the format:

Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q	Q
15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0

The 16-bit data memory locaion 'DELTA' contains the step size. DELTA has the same (integer.fraction) format as ALPHA. Every time the sine wave subroutine is called, the contents of ALPHA are incremented by the contents of DELTA. The integer portion of ALPHA (i.e., the eight MSBs) is the pointer to the sine table. However, because the table starts at address location SINE, this pointer is offset by the value for that address before the table is accessed. The eight most significant bits of ALPHA are masked when ALPHA is updated to insure that they never exceed 127. The routine returns the sine value in the data memory location 'SINA'.

For any given sampling interval, t, the frequencies which can be generated must be of the form

$$f = \frac{DELTA}{t \times 128} \text{ [Hz]} \quad \text{where t is expressed in seconds}$$

Since DELTA has a precision of eight bits to the right of the decimal place, any desired frequency ($\leq$ 1/2t [Hz]) can be approximated with an error of no more than

$$\frac{1/256}{t \times 128} \text{ [Hz]} = \frac{1}{32768 \times t} \text{ [Hz]}$$

For example, if the sampling frequency is 8 kHz, then the frequency resolution is

$$\frac{8000}{32768} \text{ Hz} = 0.25 \text{ Hz}$$

Harmonic Distortion

Due to approximations made in calculating the samples of a sine wave of frequency f, a certain amount of the "energy" of the samples' waveform will fall into other frequencies as well. These frequencies are either:

(1) Harmonic frequencies, nf, where n = 2, 3, 4, ..., or

(2) Subharmonic frequencies, nf/m, where n and m are integers.

This spurious energy results in noise which is referred to as "harmonic distortion." It is usually measured in terms of Total Harmonic Distortion (THD) which is defined as the ratio

$$THD = \frac{\text{spurious harmonic energy}}{\text{total energy of the waveform}}$$

There are two sources of error in the table lookup algorithm which cause harmonic distortion:

(1) Quantization error is introduced by representing the sine table values by 16-bit numbers.
(2) Larger errors are introduced when points between table entries are sampled. This occurs when DELTA is not an integer.

The longer the sine table is, the less significant the second error source will be. Consequently, harmonic distortion decreases with increasing table length. Furthermore, when DELTA is an integer, quantization is the only error source, and THD is extremely small regardless of table size. THD is given for several table lengths and values of DELTA in Figure 2. Note that the figures in this table only represent the THD in the digitized sine wave. If the sine wave is reconstructed using a digital-to-analog converter and analog filters, these analog devices will contribute additional distortion. (The procedure for computing THD is described in Appendix A.)

LINEAR INTERPOLATION METHOD

To decrease the harmonic distortion for a given table size, an interpolation scheme can be used to compute the sine values between table entries more accurately. Linear interpolation is the simplest method to implement. This method uses the values of two consecutive table entries as the end points of a line segment. Sample points for parameter values falling between table entries assume values on the line segment between the points. This algorithm is illustrated in Figure 3.

TABLE LENGTH: 32

DELTA	THD
2.0	0.00000024
2.25	0.00300893
2.50	0.00240751
2.75	0.00300917
3.0	0.00000024
8.25	0.00300924
11.625	0.00315807

TABLE LENGTH: 64

DELTA	THD
2.00	0.00000048
2.25	0.00075269
2.50	0.00060219
2.75	0.00075239
3.00	0.00000018
8.25	0.00075204
11.625	0.00079078

TABLE LENGTH: 128

DELTA	THD
2.00	0.00000054
2.25	0.00018859
2.50	0.00015080
2.75	0.00018835
3.00	0.00000012
8.25	0.00018889
11.625	0.00020128

Figure 2. Total Harmonic Distortion Using Direct Table Lookup



Figure 3. Linear Interpolation

This algorithm is based on the linear approximation

$$\sin(360°(I+D)/N) \cong \sin(360°I/N) + D \times \{\sin(360°(I+1)/N) - \sin(360°I/N)\}$$

$$= S[I] + D \times \{ S[I+1] - S[I] \}$$

where N is the sine table length,
I is an integer such that $0 \leq I \leq N-1$, and
D is a decimal number such that $0 \leq D < 1.0$

The value, $S[I+1] - S[I]$, is the slope of the line segment between the two sample points which bracket the value $I+D$ (i.e., $I \leq I+D < I+1$).

All the values required for this interpolation scheme are stored in the following two tables:

INDEX	ANGLE	SINE TABLE	SLOPE TABLE
0	0 X 360°/N	S[0] = sin(0°/N)	S[1] − S[0]
1	1 X 360°/N	S[1] = sin(360°/N)	S[2] − S[1]
2	2 X 360°/N	S[2] = sin(720°/N)	S[3] − S[2]
.	.	.	.
.	.	.	.
.	.	.	.
N − 2	(N − 2) X 360°/N	S[N − 2] = sin(360°(N − 2)/N)	S[N − 1] − S[N − 2]
N − 1	(N − 1) X 360°/N	S[N − 1] = sin(360°(N − 1)/N)	S[0] − S[N − 1]

TMS32010 Implementation

The sample TMS32010 implementation of this linear interpolation scheme, given in Appendix C, is an enhancement of the table lookup method. This subroutine is based on 128-entry sine and slope tables. Each time this subroutine is called, the next sample point is calculated. This subroutine uses:

(1) 276 (= 128 + 128 + 20) words of program memory space
  (128 words for sine table storage,
  128 words for slope table storage, and
  20 words for program memory)

(2) 9 words in data memory as working registers

If this program is used as a subroutine, each sample can be computed in 5.4 microseconds. However, if the code is inserted directly in line with the code of a master program, avoiding the overhead of a subroutine, a sample can be computed in only 4.6 microseconds.

Just as in the table lookup algorithm, a sine wave is generated by stepping through the sine table at a constant rate, wrapping around at the end of the table whenever 360° is exceeded. The table index is used as the angle parameter, denoted by ALPHA.

DELTA denotes the step size for this routine also. In this case, however, sample points falling between the samples in the sine table are evaluated using the linear approximation formula given above.

The values in both the sine and slope tables are calculated in the same way as they were for the table lookup program. The decimal values, +1.0 and −1.0, are represented by the two's complement hexadecimal values 4000 and C000, respectively. All hexadecimal values are rounded rather than truncated to the closest 16-bit representations to reduce quantization noise.

Because the method to compute the step size is the same as that used in the table lookup scheme, the frequency resolution will also be the same. However, because of the linear interpolation between table entries, sine values are no longer limited to the values stored in the table. This allows the error between the computed value and the actual value to be less.

Harmonic Distortion

Figure 4 lists the distortion of several sine waves generated using the TMS32010 linear interpolation routine for various table lengths and step sizes. These results clearly show that the distortion for a particular fractional step size decreases if the size of the table is increased just as in the direct table lookup case. However, for the same non-integer step size and the same table length, the distortion for the linear distortion method is much lower than that of direct table lookup.

These values were experimentally determined and the method used to compute them is given in Appendix A.

TABLE LENTH: 32

DELTA	THD
2,0	0.00000024
2.25	0.00169343
2.50	0.00135476
2.75	0.00169379
3.0	0.00000024
8.25	0.00169361
11.625	0.00177808

TABLE LENGTH: 64

DELTA	THD
2.00	0.00000048
2.25	0.00018884
2.50	0.00015055
2.75	0.00018771
3.00	0.00000018
8.25	0.00018806
11.625	0.00019815

TABLE LENGTH: 128

DELTA	THD
2.00	0.00000054
2.25	0.00000054
2.50	0.00000012
2.75	0.00000101
3.00	0.00000012
8.25	0.00000006
11.625	0.00000155

Figure 4. Harmonic Distortion Using Linear Interpolation

IMPLEMENTATION TRADE-OFFS

There are three trade-offs that must be considered when implementing the algorithms described above. They are speed, accuracy, and the size of the table ROM.

The direct table lookup method is the fastest implementation. Using a table that ranges from 0° to 360°, the routine needs only to address the table and compute the next angle. However, the table occupies more program memory space than is absolutely required.

To minimize the amount of program memory required for the sine table, one can take advantage of the symmetry of the sine function. By keeping track of the quadrant as ALPHA is increased, a table that ranges from 0° to 90° will be sufficient. This decreases the size of the table by three-fourths. However, the extra code necessary to keep track of the quadrant will increase the execution time of the routine.

If harmonic distortion is important, then some form of interpolation is needed. One can use the linear interpolation method of the second example or other approximations such as a Taylor Series or Maclaurin Series expansions carried out to the second- or third-order term, or beyond. These schemes will, however, also increase the amount of code as well as the execution time.

APPENDIX A: COMPUTATION OF TOTAL HARMONIC DISTORTION

To determine the Total Harmonic Distortion (THD) of a sampled data sine wave, the amount of energy due to frequency components other than the fundamental is divided by the total energy of the wave. This is computed from the formula:

$$THD = [E(total) - E(fundamental)] / E(total)$$

For the most accurate results, these energy terms should be calculated over a full cycle of the signal. In the case of a sine wave generated by either of the two methods, a full cycle may actually consist of several sinusoid periods. For instance, If N = table length = 128 and if DELTA = step size = 1.5, a cycle will only be completed for the smallest n for which n x 1.5 is evenly divisible by 128. This occurs for n = 256 which marks the end of the second sinusoid period.

In general, if DELTA = A/B where A and B are relatively prime integers, and N = table length, then the sequence x(n), n = 1,2,3, ... of sine-wave samples will cycle after no more than B x N points.

The amount of total "energy" in a cycle of this length is

$$E(total) = \sum_{n=0}^{BN-1} x^2(n)$$

The amount of "energy" in the fundamental frequency over this period is

$$\begin{aligned} E(fundamental) &= 1/BN\ (|X(A)|^2 + |X(BN-A)|^2) \\ &= 2/BN\ |X(A)|^2 \quad \text{for a real sequence} \end{aligned}$$

where the X(k) terms are terms of the Discrete Fourier Transform defined by the equation

$$X(k) = \sum_{n=0}^{BN-1} x(n)\exp(-j(2\pi/N)nk)$$

The values given in Figures 2 and 4 are based on actual values computed by the TMS32010 for the two sample sine-wave generator programs. The computation of THD was carried out on a VAX 11/780 using the above formulas with double-precision floating-point arithmetic.

APPENDIX B: TMS32010 TABLE LOOKUP ROUTINE

```
GENER1        320 FAMILY MACRO ASSEMBLER  2.1 83.076       17:14:48   1/18/84
                                                                  PAGE 0001

                          IDT      'GENER1'
                 **********************************************************
                 *          SINE WAVE GENERATOR                           *
                 *        DIRECT TABLE LOOKUP METHOD                      *
                 * THIS PROGRAM USES A LOOKUP TABLE OF SINE VALUES TO     *
                 * COMPUTE THE SAMPLES OF THE WAVE.  THE FREQUENCY IS     *
                 * DETERMINED BY THE SIZE BY WHICH ONE STEPS THROUGH THE  *
                 * TABLE. THE TABLE CONSISTS OF 128 ENTRIES THAT CORRESPOND *
                 * TO EQUALLY SPACED ANGLES BETWEEN 0 AND 360 DEGREES.    *
                 **********************************************************
                 *          NOTE:  Q NOTATION                             *
                 * THE TMS32010 USES FIXED-POINT TWO'S COMPLEMENT NUMBERS. *
                 * EACH 16-BIT NUMBER HAS A SIGN BIT, i INTEGER BITS, AND *
                 * (15-i) FRACTIONAL BITS.  THE VALUE AFTER THE LETTER Q  *
                 * REFERS TO THE NUMBER OF FRACTIONAL BITS THAT ARE       *
                 * REPRESENTED BY THAT NUMBER, i.e., A Q14 NUMBER IS      *
                 * CONSIDERED TO HAVE 14 FRACTIONAL BITS.                 *
                 **********************************************************
0000 F900                 B        START
0001 0083'
0002 0000  SINE           DATA     >0                *THE SINE TABLE
0003 0324                 DATA     >324              *VALUES ARE REPRESENTED IN
0004 0646                 DATA     >646              *Q14 FORMAT, i.e., THERE
0005 0964                 DATA     >964              *ARE 14 BITS AFTER THE
0006 0C7C                 DATA     >C7C              *BINARY POINT.
0007 0F8D                 DATA     >F8D
0008 1294                 DATA     >1294
0009 1590                 DATA     >1590
000A 187E                 DATA     >187E
000B 1B5D                 DATA     >1B5D
000C 1E2B                 DATA     >1E2B
000D 20E7                 DATA     >20E7
000E 238E                 DATA     >238E
000F 2620                 DATA     >2620
0010 289A                 DATA     >289A
0011 2AFB                 DATA     >2AFB
0012 2D41                 DATA     >2D41
0013 2F6C                 DATA     >2F6C
0014 3179                 DATA     >3179
0015 3368                 DATA     >3368
0016 3537                 DATA     >3537
0017 36E5                 DATA     >36E5
0018 3871                 DATA     >3871
0019 39DB                 DATA     >39DB
001A 3B21                 DATA     >3B21
001B 3C42                 DATA     >3C42
001C 3D3F                 DATA     >3D3F
001D 3E15                 DATA     >3E15
001E 3EC5                 DATA     >3EC5
001F 3F4F                 DATA     >3F4F
0020 3FB1                 DATA     >3FB1
0021 3FEC                 DATA     >3FEC
0022 4000                 DATA     >4000
0023 3FEC                 DATA     >3FEC
```

```
0024 3FB1        DATA    >3FB1
0025 3F4F        DATA    >3F4F
0026 3EC5        DATA    >3EC5
0027 3E15        DATA    >3E15
0028 3D3F        DATA    >3D3F
0029 3C42        DATA    >3C42
002A 3B21        DATA    >3B21
002B 39DB        DATA    >39DB
002C 3871        DATA    >3871
002D 36E5        DATA    >36E5
002E 3537        DATA    >3537
002F 3368        DATA    >3368
0030 3179        DATA    >3179
0031 2F6C        DATA    >2F6C
0032 2D41        DATA    >2D41
0033 2AFB        DATA    >2AFB
0034 289A        DATA    >289A
0035 2620        DATA    >2620
0036 238E        DATA    >238E
0037 20E7        DATA    >20E7
0038 1E2B        DATA    >1E2B
0039 1B5D        DATA    >1B5D
003A 187E        DATA    >187E
003B 1590        DATA    >1590
003C 1294        DATA    >1294
003D 0F8D        DATA    >F8D
003E 0C7C        DATA    >C7C
003F 0964        DATA    >964
0040 0646        DATA    >646
0041 0324        DATA    >324
0042 0000        DATA    >0
0043 FCDC        DATA    >FCDC
0044 F9BA        DATA    >F9BA
0045 F69C        DATA    >F69C
0046 F384        DATA    >F384
0047 F073        DATA    >F073
0048 ED6C        DATA    >ED6C
0049 EA70        DATA    >EA70
004A E782        DATA    >E782
004B E4A3        DATA    >E4A3
004C E1D5        DATA    >E1D5
004D DF19        DATA    >DF19
004E DC72        DATA    >DC72
004F D9E0        DATA    >D9E0
0050 D766        DATA    >D766
0051 D505        DATA    >D505
0052 D2BF        DATA    >D2BF
0053 D094        DATA    >D094
0054 CE87        DATA    >CE87
0055 CC98        DATA    >CC98
0056 CAC9        DATA    >CAC9
0057 C91B        DATA    >C91B
0058 C78F        DATA    >C78F
0059 C625        DATA    >C625
```

```
005A C4DF          DATA     >C4DF
005B C3BE          DATA     >C3BE
005C C2C1          DATA     >C2C1
005D C1EB          DATA     >C1EB
005E C13B          DATA     >C13B
005F C0B1          DATA     >C0B1
0060 C04F          DATA     >C04F
0061 C014          DATA     >C014
0062 C000          DATA     >C000
0063 C014          DATA     >C014
0064 C04F          DATA     >C04F
0065 C0B1          DATA     >C0B1
0066 C13B          DATA     >C13B
0067 C1EB          DATA     >C1EB
0068 C2C1          DATA     >C2C1
0069 C3BE          DATA     >C3BE
006A C4DF          DATA     >C4DF
006B C625          DATA     >C625
006C C78F          DATA     >C78F
006D C91B          DATA     >C91B
006E CAC9          DATA     >CAC9
006F CC98          DATA     >CC98
0070 CE87          DATA     >CE87
0071 D094          DATA     >D094
0072 D2BF          DATA     >D2BF
0073 D505          DATA     >D505
0074 D766          DATA     >D766
0075 D9E0          DATA     >D9E0
0076 DC72          DATA     >DC72
0077 DF19          DATA     >DF19
0078 E1D5          DATA     >E1D5
0079 E4A3          DATA     >E4A3
007A E782          DATA     >E782
007B EA70          DATA     >EA70
007C ED6C          DATA     >ED6C
007D F073          DATA     >F073
007E F384          DATA     >F384
007F F69C          DATA     >F69C
0080 F9BA          DATA     >F9BA
0081 FCDC          DATA     >FCDC
0082
0082 7FFF  M1      DATA     >7FFF
0083
                   ****************************
                   *DATA MEMORY LOCATIONS USED*
                   ****************************
           0000    DELTA    EQU      0
           0001    ALPHA    EQU      1
           0002    SINA     EQU      2
           0003    TEMP     EQU      3              *WORKSPACE REGISTER
           0004    MASK     EQU      4
           0005    OFSET    EQU      5
0083
```

```
               ************************************************************
               * NECESSARY INITIALIZATIONS:                               *
               * MASK    INITIALIZED TO >7FFF FOR 128 POINT TABLE         *
               * OFSET   INITIALIZED TO THE ADDRESS AT THE BEGINNING      *
               *         OF TABLE.                                        *
               * ALPHA   INITIALLY CLEARED                                *
               * DELTA   INITIALIZED TO INCREMENT VALUE USING Q8 FORMAT  *
               ************************************************************
0083 6F00      START  LDP     0                * SET DATA PAGE POINTER
0084 7E82             LACK    M1
0085 6704             TBLR    MASK
0086 7E02             LACK    SINE
0087 5005             SACL    OFSET
0088 7F89             ZAC
0089 5001             SACL    ALPHA
008A 4100             IN      DELTA,PA1                 * IN THIS EXAMPLE,
008B F800      L1     CALL    SWAVE1                    * DELTA IS INPUT
008C 008F'
               **********************************
               *       REST OF PROGRAM          *
               **********************************
008D F900             B       L1
008E 008B'
008F
               ************************************************************
               * SINE WAVE SUBROUTINE:                                    *
               * THIS ROUTINE EXTRACTS THE SINE OF AN ANGLE FROM THE      *
               * TABLE AND RETURNS THE VALUE IN THE DATA LOCATION         *
               * 'SINA'.  IT USES A FRACTIONAL STEP SIZE TO COMPUTE       *
               * THE NEXT POINT OF THE WAVE.  IT TAKES 2.6 microseconds  *
               * TO EXECUTE.                                              *
               ************************************************************
008F 2801      SWAVE1 LAC     ALPHA,8
0090 5803             SACH    TEMP             *ISOLATE INTEGER PORTION
0091 2003             LAC     TEMP
0092 0005             ADD     OFSET
0093 6702             TBLR    SINA             *SINE VALUE FROM TABLE (Q14)
0094 2001             LAC     ALPHA
0095 0000             ADD     DELTA            *COMPUTE NEXT ADDRRESS
0096 7904             AND     MASK             *MODULO 128 MASK = >7FFF
0097 5001             SACL    ALPHA            *SAVE NEXT ADDRESS
0098 7F8D             RET                      *RETURN TO MAIN PROGRAM
                      END
NO ERRORS, NO WARNINGS
```

APPENDIX C: TMS32010 LINEAR INTERPOLATION ROUTINE

```
GENER2       320 FAMILY MACRO ASSEMBLER  2.1 83.076       10:54:48    1/19/84
                                                                     PAGE 0001

                       IDT     'GENER2'
                *****************************************************
                *         SINE WAVE GENERATOR                       *
                *         LINEAR INTERPOLATION METHOD               *
                * THIS PROGRAM USES A LOOKUP TABLE OF SINE VALUES TO *
                * COMPUTE THE SAMPLES OF THE WAVE.  THE FREQUENCY IS *
                * DETERMINED BY THE SIZE BY WHICH ONE STEPS THROUGH  *
                * THE TABLE.  THE TABLE CONSISTS OF 128 ENTRIES THAT *
                * CORRESPOND TO EQUALLY SPACED ANGLES BETWEEN 0 AND  *
                * 360 DEGREES.  POINTS BETWEEN THE TABLE ENTRIES ARE *
                * APPROXIMATED USING A LINEAR APPROXIMATION,         *
                * sin(A) ~= sin(INT[A])                              *
                *           + {sin(INT[A]+1)-sin(INT[A])}xFRACT[A]   *
                * ALL THE POSSIBLE SLOPES BETWEEN ANY TWO CONSECUTIVE*
                * SINE TABLE ENTRIES ARE STORED IN A SEPARATE TABLE. *
                * THE SLOPE TABLE IS ALSO 128 ENTRIES LONG.          *
                *****************************************************
                *         NOTE:  Q NOTATION                          *
                * THE TMS32010 USES FIXED-POINT TWO'S COMPLEMENT     *
                * NUMBERS.  EACH 16-BIT NUMBER HAS A SIGN BIT, i     *
                * INTEGER BITS, AND (15-i) FRACTIONAL BITS.  THE     *
                * VALUE AFTER THE LETTER Q REFERS TO THE NUMBER OF   *
                * FRACTIONAL BITS THAT ARE REPRESENTED BY THAT       *
                * NUMBER, i.e., A Q14 NUMBER IS CONSIDERED TO HAVE   *
                * 14 FRACTIONAL BITS.                                *
                *****************************************************
0000
0000 F900              B       START
0001 0104'
                *
0002 7FFF       M1     DATA    >7FFF              *MASK VALUES
0003 0FFF       M2     DATA    >0FFF
                *
0004 0000       SINE   DATA    >0                 *THE SINE TABLE
0005 0324              DATA    >324               *VALUES ARE REPRESENTED
0006 0646              DATA    >646               *IN Q14 FORMAT, i.e.,
0007 0964              DATA    >964               *THERE ARE 14 BITS AFTER
0008 0C7C              DATA    >C7C               *THE BINARY POINT.
0009 0F8D              DATA    >F8D
000A 1294              DATA    >1294
000B 1590              DATA    >1590
000C 187E              DATA    >187E
000D 1B5D              DATA    >1B5D
000E 1E2B              DATA    >1E2B
000F 20E7              DATA    >20E7
0010 238E              DATA    >238E
0011 2620              DATA    >2620
0012 289A              DATA    >289A
0013 2AFB              DATA    >2AFB
0014 2D41              DATA    >2D41
0015 2F6C              DATA    >2F6C
0016 3179              DATA    >3179
0017 3368              DATA    >3368
0018 3537              DATA    >3537
```

```
0019 36E5        DATA    >36E5
001A 3871        DATA    >3871
001B 39DB        DATA    >39DB
001C 3B21        DATA    >3B21
001D 3C42        DATA    >3C42
001E 3D3F        DATA    >3D3F
001F 3E15        DATA    >3E15
0020 3EC5        DATA    >3EC5
0021 3F4F        DATA    >3F4F
0022 3FB1        DATA    >3FB1
0023 3FEC        DATA    >3FEC
0024 4000        DATA    >4000
0025 3FEC        DATA    >3FEC
0026 3FB1        DATA    >3FB1
0027 3F4F        DATA    >3F4F
0028 3EC5        DATA    >3EC5
0029 3E15        DATA    >3E15
002A 3D3F        DATA    >3D3F
002B 3C42        DATA    >3C42
002C 3B21        DATA    >3B21
002D 39DB        DATA    >39DB
002E 3871        DATA    >3871
002F 36E5        DATA    >36E5
0030 3537        DATA    >3537
0031 3368        DATA    >3368
0032 3179        DATA    >3179
0033 2F6C        DATA    >2F6C
0034 2D41        DATA    >2D41
0035 2AFB        DATA    >2AFB
0036 289A        DATA    >289A
0037 2620        DATA    >2620
0038 238E        DATA    >238E
0039 20E7        DATA    >20E7
003A 1E2B        DATA    >1E2B
003B 1B5D        DATA    >1B5D
003C 187E        DATA    >187E
003D 1590        DATA    >1590
003E 1294        DATA    >1294
003F 0F8D        DATA    >F8D
0040 0C7C        DATA    >C7C
0041 0964        DATA    >964
0042 0646        DATA    >646
0043 0324        DATA    >324
0044 0000        DATA    >0
0045 FCDC        DATA    >FCDC
0046 F9BA        DATA    >F9BA
0047 F69C        DATA    >F69C
0048 F384        DATA    >F384
0049 F073        DATA    >F073
004A ED6C        DATA    >ED6C
004B EA70        DATA    >EA70
004C E782        DATA    >E782
004D E4A3        DATA    >E4A3
004E E1D5        DATA    >E1D5
```

```
004F DF19          DATA    >DF19
0050 DC72          DATA    >DC72
0051 D9E0          DATA    >D9E0
0052 D766          DATA    >D766
0053 D505          DATA    >D505
0054 D2BF          DATA    >D2BF
0055 D094          DATA    >D094
0056 CE87          DATA    >CE87
0057 CC98          DATA    >CC98
0058 CAC9          DATA    >CAC9
0059 C91B          DATA    >C91B
005A C78F          DATA    >C78F
005B C625          DATA    >C625
005C C4DF          DATA    >C4DF
005D C3BE          DATA    >C3BE
005E C2C1          DATA    >C2C1
005F C1EB          DATA    >C1EB
0060 C13B          DATA    >C13B
0061 C0B1          DATA    >C0B1
0062 C04F          DATA    >C04F
0063 C014          DATA    >C014
0064 C000          DATA    >C000
0065 C014          DATA    >C014
0066 C04F          DATA    >C04F
0067 C0B1          DATA    >C0B1
0068 C13B          DATA    >C13B
0069 C1EB          DATA    >C1EB
006A C2C1          DATA    >C2C1
006B C3BE          DATA    >C3BE
006C C4DF          DATA    >C4DF
006D C625          DATA    >C625
006E C78F          DATA    >C78F
006F C91B          DATA    >C91B
0070 CAC9          DATA    >CAC9
0071 CC98          DATA    >CC98
0072 CE87          DATA    >CE87
0073 D094          DATA    >D094
0074 D2BF          DATA    >D2BF
0075 D505          DATA    >D505
0076 D766          DATA    >D766
0077 D9E0          DATA    >D9E0
0078 DC72          DATA    >DC72
0079 DF19          DATA    >DF19
007A E1D5          DATA    >E1D5
007B E4A3          DATA    >E4A3
007C E782          DATA    >E782
007D EA70          DATA    >EA70
007E ED6C          DATA    >ED6C
007F F073          DATA    >F073
0080 F384          DATA    >F384
0081 F69C          DATA    >F69C
0082 F9BA          DATA    >F9BA
0083 FCDC          DATA    >FCDC
0084 0324   TSLOPE DATA    >324            *SLOPE BETWEEN TWO
```

```
0162 0085 0322          DATA    >322          *SINE ENTRIES (Q14)
0163 0086 031E          DATA    >31E
0164 0087 0318          DATA    >318
0165 0088 0311          DATA    >311
0166 0089 0307          DATA    >307
0167 008A 02FC          DATA    >2FC
0168 008B 02EE          DATA    >2EE
0169 008C 02DF          DATA    >2DF
0170 008D 02CE          DATA    >2CE
0171 008E 02BC          DATA    >2BC
0172 008F 02A7          DATA    >2A7
0173 0090 0291          DATA    >291
0174 0091 027A          DATA    >27A
0175 0092 0261          DATA    >261
0176 0093 0246          DATA    >246
0177 0094 022B          DATA    >22B
0178 0095 020D          DATA    >20D
0179 0096 01EF          DATA    >1EF
0180 0097 01CF          DATA    >1CF
0181 0098 01AE          DATA    >1AE
0182 0099 018C          DATA    >18C
0183 009A 016A          DATA    >16A
0184 009B 0146          DATA    >146
0185 009C 0121          DATA    >121
0186 009D 00FC          DATA    >FC
0187 009E 00D6          DATA    >D6
0188 009F 00B0          DATA    >B0
0189 00A0 0089          DATA    >89
0190 00A1 0062          DATA    >62
0191 00A2 003B          DATA    >3B
0192 00A3 0014          DATA    >14
0193 00A4 FFEC          DATA    >FFEC
0194 00A5 FFC5          DATA    >FFC5
0195 00A6 FF9E          DATA    >FF9E
0196 00A7 FF77          DATA    >FF77
0197 00A8 FF50          DATA    >FF50
0198 00A9 FF2A          DATA    >FF2A
0199 00AA FF04          DATA    >FF04
0200 00AB FEDF          DATA    >FEDF
0201 00AC FEBA          DATA    >FEBA
0202 00AD FE96          DATA    >FE96
0203 00AE FE74          DATA    >FE74
0204 00AF FE52          DATA    >FE52
0205 00B0 FE31          DATA    >FE31
0206 00B1 FE11          DATA    >FE11
0207 00B2 FDF3          DATA    >FDF3
0208 00B3 FDD5          DATA    >FDD5
0209 00B4 FDBA          DATA    >FDBA
0210 00B5 FD9F          DATA    >FD9F
0211 00B6 FD86          DATA    >FD86
0212 00B7 FD6F          DATA    >FD6F
0213 00B8 FD59          DATA    >FD59
0214 00B9 FD44          DATA    >FD44
0215 00BA FD32          DATA    >FD32
```

```
0216 00BB FD21          DATA     >FD21
0217 00BC FD12          DATA     >FD12
0218 00BD FD04          DATA     >FD04
0219 00BE FCF9          DATA     >FCF9
0220 00BF FCEF          DATA     >FCEF
0221 00C0 FCE8          DATA     >FCE8
0222 00C1 FCE2          DATA     >FCE2
0223 00C2 FCDE          DATA     >FCDE
0224 00C3 FCDC          DATA     >FCDC
0225 00C4 FCDC          DATA     >FCDC
0226 00C5 FCDE          DATA     >FCDE
0227 00C6 FCE2          DATA     >FCE2
0228 00C7 FCE8          DATA     >FCE8
0229 00C8 FCEF          DATA     >FCEF
0230 00C9 FCF9          DATA     >FCF9
0231 00CA FD04          DATA     >FD04
0232 00CB FD12          DATA     >FD12
0233 00CC FD21          DATA     >FD21
0234 00CD FD32          DATA     >FD32
0235 00CE FD44          DATA     >FD44
0236 00CF FD59          DATA     >FD59
0237 00D0 FD6F          DATA     >FD6F
0238 00D1 FD86          DATA     >FD86
0239 00D2 FD9F          DATA     >FD9F
0240 00D3 FDBA          DATA     >FDBA
0241 00D4 FDD5          DATA     >FDD5
0242 00D5 FDF3          DATA     >FDF3
0243 00D6 FE11          DATA     >FE11
0244 00D7 FE31          DATA     >FE31
0245 00D8 FE52          DATA     >FE52
0246 00D9 FE74          DATA     >FE74
0247 00DA FE96          DATA     >FE96
0248 00DB FEBA          DATA     >FEBA
0249 00DC FEDF          DATA     >FEDF
0250 00DD FF04          DATA     >FF04
0251 00DE FF2A          DATA     >FF2A
0252 00DF FF50          DATA     >FF50
0253 00E0 FF77          DATA     >FF77
0254 00E1 FF9E          DATA     >FF9E
0255 00E2 FFC5          DATA     >FFC5
0256 00E3 FFEC          DATA     >FFEC
0257 00E4 0014          DATA     >14
0258 00E5 003B          DATA     >3B
0259 00E6 0062          DATA     >62
0260 00E7 0089          DATA     >89
0261 00E8 00B0          DATA     >B0
0262 00E9 00D6          DATA     >D6
0263 00EA 00FC          DATA     >FC
0264 00EB 0121          DATA     >121
0265 00EC 0146          DATA     >146
0266 00ED 016A          DATA     >16A
0267 00EE 018C          DATA     >18C
0268 00EF 01AE          DATA     >1AE
0269 00F0 01CF          DATA     >1CF
```

```
00F1 01EF        DATA    >1EF
00F2 020D        DATA    >20D
00F3 022B        DATA    >22B
00F4 0246        DATA    >246
00F5 0261        DATA    >261
00F6 027A        DATA    >27A
00F7 0291        DATA    >291
00F8 02A7        DATA    >2A7
00F9 02BC        DATA    >2BC
00FA 02CE        DATA    >2CE
00FB 02DF        DATA    >2DF
00FC 02EE        DATA    >2EE
00FD 02FC        DATA    >2FC
00FE 0307        DATA    >307
00FF 0311        DATA    >311
0100 0318        DATA    >318
0101 031E        DATA    >31E
0102 0322        DATA    >322
0103 0324        DATA    >324
0104
          ******************************
          *DATA MEMORY LOCATIONS USED *
          ******************************
     0000 DELTA  EQU    0
     0001 ALPHA  EQU    1
     0002 SLOPE  EQU    2
     0003 SINA   EQU    3
     0004 TEMP   EQU    4                *SCRATCH PAD LOCATION
     0005 MASK1  EQU    5                *MASK FOR MODULO 128 POINTER
     0006 MASK2  EQU    6                *ISOLATE FRACTIONAL PART
     0007 OFSET1 EQU    7                *ADDRESS OF SINE TABLE
     0008 OFSET2 EQU    8                *ADDRESS OF SLOPE TABLE
          ***************************************************
          * NECESSARY INITIALIZATIONS:                      *
          * MASK1   INITIALIZED TO >7FFF FOR 128 POINT TABLE *
          * MASK2   INITIALIZED TO >0FFF                    *
          * OFSET1  SET TO THE ADDRESS AT THE BEGINNING OF  *
          *         SINE TABLE                              *
          * OFSET2  SET TO THE ADDRESS AT THE BEGINNING OF  *
          *         SLOPE TABLE WITH RESPECT TO SINE TABLE  *
          * ALPHA   INITIALLY CLEARED                       *
          * DELTA   INITIALIZED TO INCREMENT VALUE USING    *
          *         Q8 FORMAT                               *
          ***************************************************
0104
0104 6F00 START  LDP    0                *SET DATA PAGE POINTER.
0105 7E04        LACK   SINE             *SINE TABLE ADDRESS.
0106 5007        SACL   OFSET1
0107 7E84        LACK   TSLOPE           *SLOPE TABLE ADDRESS.
0108 1007        SUB    OFSET1
0109 5008        SACL   OFSET2
010A 7E02        LACK   M1               *RETRIEVE MASK1.
010B 6705        TBLR   MASK1
010C 7E03        LACK   M2               *RETRIEVE MASK2.
```

```
010D 6706         TBLR   MASK2
010E 7F89         ZAC
010F 5001         SACL   ALPHA          *START ANGLE AT ZERO.
0110 4100         IN     DELTA,PA1      *FOR THIS EXAMPLE,
0111 F800  L1     CALL   SWAVE2         *DELTA IS INPUT.
0112 0115'
           *****************************
           *      REST OF PROGRAM      *
           *****************************
0113 F900         B      L1
0114 0111'
0115
0115
           *****************************************************
           * SINE WAVE SUBROUTINE:                             *
           * THIS ROUTINE COMPUTES THE SINE OF AN ANGLE AND    *
           * RETURNS THE VALUE IN DATA LOCATION 'SINA'.  IT USES *
           * A FRACTIONAL STEP SIZE TO AUTOMATICALLY COMPUTE THE *
           * ADDRESS OF THE NEXT POINT OF THE SINE WAVE. IT    *
           * TAKES 5.0 microseconds TO EXECUTE.                *
           *****************************************************
0115 2801  SWAVE2 LAC    ALPHA,8
0116 5804         SACH   TEMP           *ISOLATE INTEGER PORTION
0117 2004         LAC    TEMP
0118 0007         ADD    OFSET1
0119 6703         TBLR   SINA           *GET CLOSEST LOWER SINE-
011A 0008         ADD    OFSET2         *TABLE ENTRY (b).
011B 6702         TBLR   SLOPE          *SLOPE FOR CORRESPONDING INTERVA
011C 2401         LAC    ALPHA,4        *ISOLATE FRACTIONAL PORTION
011D 7906         AND    MASK2          *AND MAKE Q8 NUMBER INTO Q12.
011E 5004         SACL   TEMP
011F 6A04         LT     TEMP
0120 6D02         MPY    SLOPE          * mx
0121 7F8E         PAC
0122 0C03         ADD    SINA,12
0123 5C03         SACH   SINA,4         * y = mx + b.
0124 2001         LAC    ALPHA          *COMPUTE ADDRESS OF NEXT POINT.
0125 0000         ADD    DELTA          *OF WAVE.
0126 7905         AND    MASK1          *CLEAR MOST SIGNIFICANT BIT.
0127 5001         SACL   ALPHA          *SAVE NEXT ADDRESS.
0128 7F8D         RET                   *RETURN TO MAIN PROGRAM.
                  END
NO ERRORS, NO WARNINGS
```